# $Id: PKGBUILD 184401 2013-05-06 19:38:14Z foutrelis $
# Maintainer: Giovanni Scafora <giovanni@archlinux.org>
# Contributor: Tom Newsom <Jeepster@gmx.co.uk>

pkgname=lbreakout2
pkgver=2.6.4
pkgrel=2
pkgdesc="A breakout game with nice effects, graphics, and sounds"
arch=('i686' 'x86_64')
url="http://lgames.sourceforge.net"
license=('GPL')
depends=('sdl_mixer' 'libpng')
backup=('var/games/lbreakout2/lbreakout2.hscr')
source=("http://downloads.sourceforge.net/lgames/${pkgname}-${pkgver/_/-}.tar.gz"
        'lbreakout2.desktop'
	'lbreakout2.png')
md5sums=('4849af72225a304267eb71a9c100cd48'
         '14cd34a72a8f4e3b8f385f9023625241'
         '94e2fa393eb2a7d4494641dec580573b')

build() {
  cd ${srcdir}/${pkgname}-${pkgver/_/-}

  ./configure --prefix=/usr \
              --localstatedir=/var/games/lbreakout2 
  make
}

package() {
  cd ${srcdir}/${pkgname}-${pkgver/_/-}

  make DESTDIR=${pkgdir}/ doc_dir=/usr/share/doc install
  
  chmod -R 755 ${pkgdir}/usr
  chown root:games ${pkgdir}/var/games
  chmod 775 ${pkgdir}/var/games
  chown root:games ${pkgdir}/var/games/lbreakout2
  chmod 755 ${pkgdir}/var/games/lbreakout2
  chown root:games ${pkgdir}/var/games/lbreakout2/lbreakout2.hscr
  chmod 775 ${pkgdir}/var/games/lbreakout2/lbreakout2.hscr

  install -Dm644 ${srcdir}/lbreakout2.png ${pkgdir}/usr/share/pixmaps/lbreakout2.png
  install -Dm644 ${srcdir}/lbreakout2.desktop ${pkgdir}/usr/share/applications/lbreakout2.desktop
}
